Rules for integrands of the form $(g \operatorname{Sin}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Sin}[e+f x])^n$

1. $\int \frac{(g \operatorname{Sin}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m}{c+d \operatorname{Sin}[e+f x]} dx$ when $b c-a d \neq 0$

   1. $\int \frac{(g \operatorname{Sin}[e+f x])^p \sqrt{a+b \operatorname{Sin}[e+f x]}}{c+d \operatorname{Sin}[e+f x]} dx$ when $b c-a d \neq 0$

      1. $\int \frac{\sqrt{g \operatorname{Sin}[e+f x]} \sqrt{a+b \operatorname{Sin}[e+f x]}}{c+d \operatorname{Sin}[e+f x]} dx$ when $b c-a d \neq 0$

         1: $\int \frac{\sqrt{g \operatorname{Sin}[e+f x]} \sqrt{a+b \operatorname{Sin}[e+f x]}}{c+d \operatorname{Sin}[e+f x]} dx$ when $b c-a d \neq 0 \wedge \left(a^2-b^2=0 \vee c^2-d^2=0\right)$

Derivation: Algebraic expansion

Basis: $\frac{\sqrt{g z}}{c+d z} = \frac{g}{d \sqrt{g z}} - \frac{c g}{d \sqrt{g z} (c+d z)}$

Rule: If $b c-a d \neq 0 \wedge \left(a^2-b^2=0 \vee c^2-d^2=0\right)$, then

$$\int \frac{\sqrt{g \operatorname{Sin}[e+f x]} \sqrt{a+b \operatorname{Sin}[e+f x]}}{c+d \operatorname{Sin}[e+f x]} dx \rightarrow \frac{g}{d} \int \frac{\sqrt{a+b \operatorname{Sin}[e+f x]}}{\sqrt{g \operatorname{Sin}[e+f x]}} dx - \frac{c g}{d} \int \frac{\sqrt{a+b \operatorname{Sin}[e+f x]}}{\sqrt{g \operatorname{Sin}[e+f x]} (c+d \operatorname{Sin}[e+f x])} dx$$

Program code:

```
Int[Sqrt[g_.*sin[e_.+f_.*x_]]*Sqrt[a_+b_.*sin[e_.+f_.*x_]]/(c_+d_.*sin[e_.+f_.*x_]),x_Symbol] :=
  g/d*Int[Sqrt[a+b*Sin[e+f*x]]/Sqrt[g*Sin[e+f*x]],x] -
  c*g/d*Int[Sqrt[a+b*Sin[e+f*x]]/(Sqrt[g*Sin[e+f*x]]*(c+d*Sin[e+f*x])),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b*c-a*d,0] && (EqQ[a^2-b^2,0] || EqQ[c^2-d^2,0])
```

2: $\int \frac{\sqrt{g \operatorname{Sin}[e+f x]} \sqrt{a+b \operatorname{Sin}[e+f x]}}{c+d \operatorname{Sin}[e+f x]} dx$ when $b c-a d \neq 0 \wedge a^2-b^2 \neq 0 \wedge c^2-d^2 \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{\sqrt{a+b z}}{c+d z} == \frac{b}{d \sqrt{a+b z}} - \frac{b c-a d}{d \sqrt{a+b z} (c+d z)}$

- Rule: If $b c - a d \neq 0 \land a^2 - b^2 \neq 0 \land c^2 - d^2 \neq 0$, then

$$\int \frac{\sqrt{g \sin[e+f x]} \sqrt{a+b \sin[e+f x]}}{c+d \sin[e+f x]} dx \longrightarrow \frac{b}{d} \int \frac{\sqrt{g \sin[e+f x]}}{\sqrt{a+b \sin[e+f x]}} dx - \frac{b c-a d}{d} \int \frac{\sqrt{g \sin[e+f x]}}{\sqrt{a+b \sin[e+f x]} (c+d \sin[e+f x])} dx$$

- Program code:

```
Int[Sqrt[g_.*sin[e_.+f_.*x_]]*Sqrt[a_+b_.*sin[e_.+f_.*x_]]/(c_+d_.*sin[e_.+f_.*x_]),x_Symbol] :=
  b/d*Int[Sqrt[g*Sin[e+f*x]]/Sqrt[a+b*Sin[e+f*x]],x] -
  (b*c-a*d)/d*Int[Sqrt[g*Sin[e+f*x]]/(Sqrt[a+b*Sin[e+f*x]]*(c+d*Sin[e+f*x])),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

2. $\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\sqrt{g\,\text{Sin}[e+f\,x]}\,(c+d\,\text{Sin}[e+f\,x])}\,dx$ when $b\,c-a\,d\neq 0$

1: $\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\sqrt{g\,\text{Sin}[e+f\,x]}\,(c+d\,\text{Sin}[e+f\,x])}\,dx$ when $b\,c-a\,d\neq 0\wedge a^2-b^2==0$

Derivation: Integration by substitution

Basis: If $a^2-b^2==0$, then $\frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\sqrt{g\,\text{Sin}[e+f\,x]}\,(c+d\,\text{Sin}[e+f\,x])}==$

$$-\frac{2\,b}{f}\,\text{Subst}\left[\frac{1}{b\,c+a\,d+c\,g\,x^2},\,x,\,\frac{b\,\text{Cos}[e+f\,x]}{\sqrt{g\,\text{Sin}[e+f\,x]}\,\sqrt{a+b\,\text{Sin}[e+f\,x]}}\right]\,\partial_x\,\frac{b\,\text{Cos}[e+f\,x]}{\sqrt{g\,\text{Sin}[e+f\,x]}\,\sqrt{a+b\,\text{Sin}[e+f\,x]}}$$

Rule: If $b\,c-a\,d\neq 0\wedge a^2-b^2==0$, then

$$\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\sqrt{g\,\text{Sin}[e+f\,x]}\,(c+d\,\text{Sin}[e+f\,x])}\,dx\,\rightarrow\,-\frac{2\,b}{f}\,\text{Subst}\left[\int\frac{1}{b\,c+a\,d+c\,g\,x^2}\,dx,\,x,\,\frac{b\,\text{Cos}[e+f\,x]}{\sqrt{g\,\text{Sin}[e+f\,x]}\,\sqrt{a+b\,\text{Sin}[e+f\,x]}}\right]$$

Program code:

```
Int[Sqrt[a_+b_.*sin[e_.+f_.*x_]]/(Sqrt[g_.*sin[e_.+f_.*x_]]*(c_+d_.*sin[e_.+f_.*x_])),x_Symbol] :=
  -2*b/f*Subst[Int[1/(b*c+a*d+c*g*x^2),x],x,b*Cos[e+f*x]/(Sqrt[g*Sin[e+f*x]]*Sqrt[a+b*Sin[e+f*x]])] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0]
```

2. $\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\sqrt{g\,\text{Sin}[e+f\,x]}\,(c+d\,\text{Sin}[e+f\,x])}\,dx$ when $b\,c-a\,d\neq 0\wedge a^2-b^2\neq 0$

1. $\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\sqrt{g\,\text{Sin}[e+f\,x]}\,(c+d\,\text{Sin}[e+f\,x])}\,dx$ when $b\,c-a\,d\neq 0\wedge a^2-b^2\neq 0\wedge c^2-d^2==0$

1: $\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\sqrt{\text{Sin}[e+f\,x]}\,(c+c\,\text{Sin}[e+f\,x])}\,dx$ when $a^2-b^2>0\land b>0$

Basis: If $b-a>0\land b>0$, then $\sqrt{a+b\,z}==\sqrt{1+z}\,\sqrt{\frac{a+b\,z}{1+z}}$

Rule: If $a^2-b^2>0\land b>0$, then

$$\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\sqrt{\text{Sin}[e+f\,x]}\,(c+c\,\text{Sin}[e+f\,x])}\,dx\to-\frac{\sqrt{a+b}}{c\,f}\,\text{EllipticE}\left[\text{ArcSin}\left[\frac{\text{Cos}[e+f\,x]}{1+\text{Sin}[e+f\,x]}\right],-\frac{a-b}{a+b}\right]$$

Program code:

```
Int[Sqrt[a_+b_.*sin[e_.+f_.*x_]]/(Sqrt[sin[e_.+f_.*x_]]*(c_+d_.*sin[e_.+f_.*x_])),x_Symbol] :=
  -Sqrt[a+b]/(c*f)*EllipticE[ArcSin[Cos[e+f*x]/(1+Sin[e+f*x])],-(a-b)/(a+b)] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[d,c] && GtQ[b^2-a^2,0] && GtQ[b,0]
```

2: $\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\sqrt{g\,\text{Sin}[e+f\,x]}\,(c+d\,\text{Sin}[e+f\,x])}\,dx$ when $b\,c-a\,d\neq0\land a^2-b^2\neq0\land c^2-d^2==0$

Rule: If $b\,c-a\,d\neq0\land a^2-b^2\neq0\land c^2-d^2==0$, then

$$\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\sqrt{g\,\text{Sin}[e+f\,x]}\,(c+d\,\text{Sin}[e+f\,x])}\,dx\to-\frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{\frac{d\,\text{Sin}[e+f\,x]}{c+d\,\text{Sin}[e+f\,x]}}}{d\,f\,\sqrt{g\,\text{Sin}[e+f\,x]}\,\sqrt{\frac{c^2\,(a+b\,\text{Sin}[e+f\,x])}{(a\,c+b\,d)\,(c+d\,\text{Sin}[e+f\,x])}}}\,\text{EllipticE}\left[\text{ArcSin}\left[\frac{c\,\text{Cos}[e+f\,x]}{c+d\,\text{Sin}[e+f\,x]}\right],\frac{b\,c-a\,d}{b\,c+a\,d}\right]$$

Program code:

```
Int[Sqrt[a_+b_.*sin[e_.+f_.*x_]]/(Sqrt[g_.*sin[e_.+f_.*x_]]*(c_+d_.*sin[e_.+f_.*x_])),x_Symbol] :=
  -Sqrt[a+b*Sin[e+f*x]]*Sqrt[d*Sin[e+f*x]/(c+d*Sin[e+f*x])]/
    (d*f*Sqrt[g*Sin[e+f*x]]*Sqrt[c^2*(a+b*Sin[e+f*x])/((a*c+b*d)*(c+d*Sin[e+f*x]))])*
    EllipticE[ArcSin[c*Cos[e+f*x]/(c+d*Sin[e+f*x])],(b*c-a*d)/(b*c+a*d)] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && EqQ[c^2-d^2,0]
```

2: $\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\sqrt{g\,\text{Sin}[e+f\,x]}\,(c+d\,\text{Sin}[e+f\,x])}\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,c^2-d^2\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{\sqrt{a+b\,z}}{\sqrt{g\,z}\,(c+d\,z)} == \frac{a}{c\,\sqrt{g\,z}\,\sqrt{a+b\,z}} + \frac{(b\,c-a\,d)\,\sqrt{g\,z}}{c\,g\,\sqrt{a+b\,z}\,(c+d\,z)}$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,c^2-d^2\neq 0$, then

$$\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\sqrt{g\,\text{Sin}[e+f\,x]}\,(c+d\,\text{Sin}[e+f\,x])}\,dx \rightarrow \frac{a}{c}\int \frac{1}{\sqrt{g\,\text{Sin}[e+f\,x]}\,\sqrt{a+b\,\text{Sin}[e+f\,x]}}\,dx + \frac{b\,c-a\,d}{c\,g}\int \frac{\sqrt{g\,\text{Sin}[e+f\,x]}}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,(c+d\,\text{Sin}[e+f\,x])}\,dx$$

Program code:

```
Int[Sqrt[a_+b_.*sin[e_.+f_.*x_]]/(Sqrt[g_.*sin[e_.+f_.*x_]]*(c_+d_.*sin[e_.+f_.*x_])),x_Symbol] :=
  a/c*Int[1/(Sqrt[g*Sin[e+f*x]]*Sqrt[a+b*Sin[e+f*x]]),x] +
  (b*c-a*d)/(c*g)*Int[Sqrt[g*Sin[e+f*x]]/(Sqrt[a+b*Sin[e+f*x]]*(c+d*Sin[e+f*x])),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

3. $\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\text{Sin}[e+f\,x]\,(c+d\,\text{Sin}[e+f\,x])}\,dx$ when $b\,c-a\,d\neq 0$

1: $\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\text{Sin}[e+f\,x]\,(c+d\,\text{Sin}[e+f\,x])}\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0$

Derivation: Algebraic expansion

Basis: $\frac{1}{z\,(c+d\,z)} == \frac{1}{c\,z} - \frac{d}{c\,(c+d\,z)}$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0$, then

$$\int \frac{\sqrt{a + b \operatorname{Sin}[e + f x]}}{\operatorname{Sin}[e + f x] \, (c + d \operatorname{Sin}[e + f x])} \, dx \longrightarrow \frac{1}{c} \int \frac{\sqrt{a + b \operatorname{Sin}[e + f x]}}{\operatorname{Sin}[e + f x]} \, dx - \frac{d}{c} \int \frac{\sqrt{a + b \operatorname{Sin}[e + f x]}}{c + d \operatorname{Sin}[e + f x]} \, dx$$

Program code:

```
Int[Sqrt[a_+b_.*sin[e_.+f_.*x_]]/(sin[e_.+f_.*x_]*(c_+d_.*sin[e_.+f_.*x_])),x_Symbol] :=
  1/c*Int[Sqrt[a+b*Sin[e+f*x]]/Sin[e+f*x],x] -
  d/c*Int[Sqrt[a+b*Sin[e+f*x]]/(c+d*Sin[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0]
```

2: $\int \frac{\sqrt{a + b \operatorname{Sin}[e + f x]}}{\operatorname{Sin}[e + f x] \, (c + d \operatorname{Sin}[e + f x])} \, dx$ when $b c - a d \neq 0 \land a^2 - b^2 \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{\sqrt{a+b z}}{z (c+d z)} == \frac{a}{c z \sqrt{a+b z}} + \frac{b c-a d}{c \sqrt{a+b z} (c+d z)}$

Rule: If $b c - a d \neq 0 \land a^2 - b^2 \neq 0$, then

$$\int \frac{\sqrt{a + b \operatorname{Sin}[e + f x]}}{\operatorname{Sin}[e + f x] \, (c + d \operatorname{Sin}[e + f x])} \, dx \longrightarrow \frac{a}{c} \int \frac{1}{\operatorname{Sin}[e + f x] \sqrt{a + b \operatorname{Sin}[e + f x]}} \, dx + \frac{b c - a d}{c} \int \frac{1}{\sqrt{a + b \operatorname{Sin}[e + f x]} \, (c + d \operatorname{Sin}[e + f x])} \, dx$$

Program code:

```
Int[Sqrt[a_+b_.*sin[e_.+f_.*x_]]/(sin[e_.+f_.*x_]*(c_+d_.*sin[e_.+f_.*x_])),x_Symbol] :=
  a/c*Int[1/(Sin[e+f*x]*Sqrt[a+b*Sin[e+f*x]]),x] +
  (b*c-a*d)/c*Int[1/(Sqrt[a+b*Sin[e+f*x]]*(c+d*Sin[e+f*x])),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0]
```

2. $\int \frac{(g \operatorname{Sin}[e + f x])^p}{\sqrt{a + b \operatorname{Sin}[e + f x]} \, (c + d \operatorname{Sin}[e + f x])} \, dx$ when $b c - a d \neq 0$

1. $\int \frac{\sqrt{g \operatorname{Sin}[e+f x]}}{\sqrt{a+b \operatorname{Sin}[e+f x]}\,(c+d \operatorname{Sin}[e+f x])} \, dx$ when $b c-a d \neq 0$

1: $\int \frac{\sqrt{g \operatorname{Sin}[e+f x]}}{\sqrt{a+b \operatorname{Sin}[e+f x]}\,(c+d \operatorname{Sin}[e+f x])} \, dx$ when $b c-a d \neq 0 \land \left(a^2-b^2=0 \lor c^2-d^2=0\right)$

Derivation: Algebraic expansion

Basis: $\frac{\sqrt{g z}}{\sqrt{a+b z}\,(c+d z)} = -\frac{a g}{(b c-a d)\sqrt{g z}\,\sqrt{a+b z}}+\frac{c g \sqrt{a+b z}}{(b c-a d)\sqrt{g z}\,(c+d z)}$

Rule: If $b c-a d \neq 0 \land \left(a^2-b^2=0 \lor c^2-d^2=0\right)$, then

$$\int \frac{\sqrt{g \operatorname{Sin}[e+f x]}}{\sqrt{a+b \operatorname{Sin}[e+f x]}\,(c+d \operatorname{Sin}[e+f x])} \, dx \rightarrow$$

$$-\frac{a g}{b c-a d} \int \frac{1}{\sqrt{g \operatorname{Sin}[e+f x]}\,\sqrt{a+b \operatorname{Sin}[e+f x]}} \, dx+\frac{c g}{b c-a d} \int \frac{\sqrt{a+b \operatorname{Sin}[e+f x]}}{\sqrt{g \operatorname{Sin}[e+f x]}\,(c+d \operatorname{Sin}[e+f x])} \, dx$$

Program code:

```
Int[Sqrt[g_.*sin[e_.+f_.*x_]]/(Sqrt[a_+b_.*sin[e_.+f_.*x_]]*(c_+d_.*sin[e_.+f_.*x_])),x_Symbol] :=
  -a*g/(b*c-a*d)*Int[1/(Sqrt[g*Sin[e+f*x]]*Sqrt[a+b*Sin[e+f*x]]),x] +
  c*g/(b*c-a*d)*Int[Sqrt[a+b*Sin[e+f*x]]/(Sqrt[g*Sin[e+f*x]]*(c+d*Sin[e+f*x])),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b*c-a*d,0] && (EqQ[a^2-b^2,0] || EqQ[c^2-d^2,0])
```

2: $\int \frac{\sqrt{g \operatorname{Sin}[e+f x]}}{\sqrt{a+b \operatorname{Sin}[e+f x]}\,(c+d \operatorname{Sin}[e+f x])} \, dx$ when $b c-a d \neq 0 \land a^2-b^2 \neq 0 \land c^2-d^2 \neq 0$

Rule: If $b c-a d \neq 0 \land a^2-b^2 \neq 0 \land c^2-d^2 \neq 0$, then

$$\int \frac{\sqrt{g \operatorname{Sin}[e+f x]}}{\sqrt{a+b \operatorname{Sin}[e+f x]}\,(c+d \operatorname{Sin}[e+f x])} \, dx \rightarrow$$

$$\frac{2\sqrt{-\text{Cot}[e+f x]^2}\,\sqrt{g\,\text{Sin}[e+f x]}}{f\,(c+d)\,\text{Cot}[e+f x]\,\sqrt{a+b\,\text{Sin}[e+f x]}}\sqrt{\frac{b+a\,\text{Csc}[e+f x]}{a+b}}\;\text{EllipticPi}\left[\frac{2 c}{c+d},\,\text{ArcSin}\left[\frac{\sqrt{1-\text{Csc}[e+f x]}}{\sqrt{2}}\right],\,\frac{2 a}{a+b}\right]$$

–

Program code:

```
Int[Sqrt[g_.*sin[e_.+f_.*x_]]/(Sqrt[a_+b_.*sin[e_.+f_.*x_]]*(c_+d_.*sin[e_.+f_.*x_])),x_Symbol] :=
  2*Sqrt[-Cot[e+f*x]^2]*Sqrt[g*Sin[e+f*x]]/(f*(c+d)*Cot[e+f*x]*Sqrt[a+b*Sin[e+f*x]])*Sqrt[(b+a*Csc[e+f*x])/(a+b)]*
    EllipticPi[2*c/(c+d),ArcSin[Sqrt[1-Csc[e+f*x]]/Sqrt[2]],2*a/(a+b)] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

2. $$\int \frac{1}{\sqrt{g \operatorname{Sin}[e+f x]}\ \sqrt{a+b \operatorname{Sin}[e+f x]}\ (c+d \operatorname{Sin}[e+f x])}\, dx \text{ when } b c-a d\neq 0$$

1: $$\int \frac{1}{\sqrt{g \operatorname{Sin}[e+f x]}\ \sqrt{a+b \operatorname{Sin}[e+f x]}\ (c+d \operatorname{Sin}[e+f x])}\, dx \text{ when } b c-a d\neq 0 \land \left(a^2-b^2==0 \lor c^2-d^2==0\right)$$

– Derivation: Algebraic expansion

Basis: $\frac{1}{\sqrt{a+b z}\ (c+d z)}==\frac{b}{(b c-a d)\ \sqrt{a+b z}}-\frac{d\sqrt{a+b z}}{(b c-a d)\ (c+d z)}$

– Rule: If $b c-a d\neq 0 \land \left(a^2-b^2==0 \lor c^2-d^2==0\right)$, then

$$\int \frac{1}{\sqrt{g \operatorname{Sin}[e+f x]}\ \sqrt{a+b \operatorname{Sin}[e+f x]}\ (c+d \operatorname{Sin}[e+f x])}\, dx \to$$

$$\frac{b}{b c-a d}\int \frac{1}{\sqrt{g \operatorname{Sin}[e+f x]}\ \sqrt{a+b \operatorname{Sin}[e+f x]}}\, dx-\frac{d}{b c-a d}\int \frac{\sqrt{a+b \operatorname{Sin}[e+f x]}}{\sqrt{g \operatorname{Sin}[e+f x]}\ (c+d \operatorname{Sin}[e+f x])}\, dx$$

– Program code:

```
Int[1/(Sqrt[g_.*sin[e_.+f_.*x_]]*Sqrt[a_+b_.*sin[e_.+f_.*x_]]*(c_+d_.*sin[e_.+f_.*x_])),x_Symbol] :=
  b/(b*c-a*d)*Int[1/(Sqrt[g*Sin[e+f*x]]*Sqrt[a+b*Sin[e+f*x]]),x] -
  d/(b*c-a*d)*Int[Sqrt[a+b*Sin[e+f*x]]/(Sqrt[g*Sin[e+f*x]]*(c+d*Sin[e+f*x])),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b*c-a*d,0] && (EqQ[a^2-b^2,0] || EqQ[c^2-d^2,0])
```

2: $\int \frac{1}{\sqrt{g\,\text{Sin}[e+f\,x]}\,\sqrt{a+b\,\text{Sin}[e+f\,x]}\,(c+d\,\text{Sin}[e+f\,x])}\,dx$ when $b\,c-a\,d\neq 0\land a^2-b^2\neq 0\land c^2-d^2\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{1}{\sqrt{g\,z}\,\sqrt{a+b\,z}\,(c+d\,z)}==\frac{1}{c\,\sqrt{g\,z}\,\sqrt{a+b\,z}}-\frac{d\,\sqrt{g\,z}}{c\,g\,\sqrt{a+b\,z}\,(c+d\,z)}$

Rule: If $b\,c-a\,d\neq 0\land a^2-b^2\neq 0\land c^2-d^2\neq 0$, then

$$\int \frac{1}{\sqrt{g\,\text{Sin}[e+f\,x]}\,\sqrt{a+b\,\text{Sin}[e+f\,x]}\,(c+d\,\text{Sin}[e+f\,x])}\,dx\to\frac{1}{c}\int \frac{1}{\sqrt{g\,\text{Sin}[e+f\,x]}\,\sqrt{a+b\,\text{Sin}[e+f\,x]}}\,dx-\frac{d}{c\,g}\int \frac{\sqrt{g\,\text{Sin}[e+f\,x]}}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,(c+d\,\text{Sin}[e+f\,x])}\,dx$$

Program code:

```
Int[1/(Sqrt[g_.*sin[e_.+f_.*x_]]*Sqrt[a_+b_.*sin[e_.+f_.*x_]]*(c_+d_.*sin[e_.+f_.*x_])),x_Symbol] :=
  1/c*Int[1/(Sqrt[g*Sin[e+f*x]]*Sqrt[a+b*Sin[e+f*x]]),x] -
  d/(c*g)*Int[Sqrt[g*Sin[e+f*x]]/(Sqrt[a+b*Sin[e+f*x]]*(c+d*Sin[e+f*x])),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

3. $\int \frac{1}{\text{Sin}[e+f\,x]\,\sqrt{a+b\,\text{Sin}[e+f\,x]}\,(c+d\,\text{Sin}[e+f\,x])}\,dx$ when $b\,c-a\,d\neq 0$

1: $\int \frac{1}{\text{Sin}[e+f\,x]\,\sqrt{a+b\,\text{Sin}[e+f\,x]}\,(c+d\,\text{Sin}[e+f\,x])}\,dx$ when $b\,c-a\,d\neq 0\land a^2-b^2==0$

Derivation: Algebraic expansion

Basis: $\frac{1}{z\,\sqrt{a+b\,z}\,(c+d\,z)}==\frac{b\,c-a\,d-b\,d\,z}{c\,(b\,c-a\,d)\,z\,\sqrt{a+b\,z}}+\frac{d^2\,\sqrt{a+b\,z}}{c\,(b\,c-a\,d)\,(c+d\,z)}$

Rule: If $b\,c-a\,d\neq 0\land a^2-b^2==0$, then

$$\int \frac{1}{\text{Sin}[e+f x]\sqrt{a+b\,\text{Sin}[e+f x]}\;(c+d\,\text{Sin}[e+f x])}\,dx \to \frac{d^2}{c\,(b\,c-a\,d)}\int \frac{\sqrt{a+b\,\text{Sin}[e+f x]}}{c+d\,\text{Sin}[e+f x]}\,dx + \frac{1}{c\,(b\,c-a\,d)}\int \frac{b\,c-a\,d-b\,d\,\text{Sin}[e+f x]}{\text{Sin}[e+f x]\sqrt{a+b\,\text{Sin}[e+f x]}}\,dx$$

Program code:

```
Int[1/(sin[e_.+f_.*x_]*Sqrt[a_+b_.*sin[e_.+f_.*x_]]*(c_+d_.*sin[e_.+f_.*x_])),x_Symbol] :=
  d^2/(c*(b*c-a*d))*Int[Sqrt[a+b*Sin[e+f*x]]/(c+d*Sin[e+f*x]),x] +
  1/(c*(b*c-a*d))*Int[(b*c-a*d-b*d*Sin[e+f*x])/(Sin[e+f*x]*Sqrt[a+b*Sin[e+f*x]]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0]
```

2: $\int \frac{1}{\text{Sin}[e+f x]\sqrt{a+b\,\text{Sin}[e+f x]}\;(c+d\,\text{Sin}[e+f x])}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{1}{z\,(c+d\,z)} == \frac{1}{c\,z} - \frac{d}{c\,(c+d\,z)}$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0$, then

$$\int \frac{1}{\text{Sin}[e+f x]\sqrt{a+b\,\text{Sin}[e+f x]}\;(c+d\,\text{Sin}[e+f x])}\,dx \to \frac{1}{c}\int \frac{1}{\text{Sin}[e+f x]\sqrt{a+b\,\text{Sin}[e+f x]}}\,dx - \frac{d}{c}\int \frac{1}{\sqrt{a+b\,\text{Sin}[e+f x]}\;(c+d\,\text{Sin}[e+f x])}\,dx$$

Program code:

```
Int[1/(sin[e_.+f_.*x_]*Sqrt[a_+b_.*sin[e_.+f_.*x_]]*(c_+d_.*sin[e_.+f_.*x_])),x_Symbol] :=
  1/c*Int[1/(Sin[e+f*x]*Sqrt[a+b*Sin[e+f*x]]),x] - d/c*Int[1/(Sqrt[a+b*Sin[e+f*x]]*(c+d*Sin[e+f*x])),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0]
```

2. $\int \frac{(a+b\,\text{Sin}[e+f x])^m\,(c+d\,\text{Sin}[e+f x])^n}{\text{Sin}[e+f x]}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,m^2 == n^2 == \frac{1}{4}$

1. $\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\text{Sin}[e+f\,x]\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0$

   1. $\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\text{Sin}[e+f\,x]\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2=0$

      1: $\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\text{Sin}[e+f\,x]\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2=0\,\land\,b\,c+a\,d=0$

–

Derivation: Algebraic expansion

Basis: $\frac{1}{z\,\sqrt{c+d\,z}} = -\frac{d}{c\,\sqrt{c+d\,z}} + \frac{\sqrt{c+d\,z}}{c\,z}$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2-b^2=0\,\land\,b\,c+a\,d=0$, then

$$\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\text{Sin}[e+f\,x]\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx \to -\frac{d}{c}\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx + \frac{1}{c}\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}{\text{Sin}[e+f\,x]}\,dx$$

Program code:

```
Int[Sqrt[a_+b_.*sin[e_.+f_.*x_]]/(sin[e_.+f_.*x_]*Sqrt[c_+d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  -d/c*Int[Sqrt[a+b*Sin[e+f*x]]/Sqrt[c+d*Sin[e+f*x]],x] +
  1/c*Int[Sqrt[a+b*Sin[e+f*x]]*Sqrt[c+d*Sin[e+f*x]]/Sin[e+f*x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && EqQ[b*c+a*d,0]
```

2: $\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\text{Sin}[e+f\,x]\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,b\,c+a\,d\neq 0$

Derivation: Integration by substitution

Basis: If $a^2-b^2==0\,\land\,c^2-d^2\neq 0$, then

$$\frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\text{Sin}[e+f\,x]\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}==-\frac{2\,a}{f}\,\text{Subst}\left[\frac{1}{1-a\,c\,x^2},\,x,\,\frac{\text{Cos}[e+f\,x]}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}\right]\,\partial_x\,\frac{\text{Cos}[e+f\,x]}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}$$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,b\,c+a\,d\neq 0$, then

$$\int\frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\text{Sin}[e+f\,x]\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx\,\rightarrow\,-\frac{2\,a}{f}\,\text{Subst}\left[\int\frac{1}{1-a\,c\,x^2}\,dx,\,x,\,\frac{\text{Cos}[e+f\,x]}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}\right]$$

Program code:

```
Int[Sqrt[a_+b_.*sin[e_.+f_.*x_]]/(sin[e_.+f_.*x_]*Sqrt[c_+d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  -2*a/f*Subst[Int[1/(1-a*c*x^2),x],x,Cos[e+f*x]/(Sqrt[a+b*Sin[e+f*x]]*Sqrt[c+d*Sin[e+f*x]])] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[b*c+a*d,0]
```

2. $\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\text{Sin}[e+f\,x]\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0$

1: $\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\text{Sin}[e+f\,x]\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,c^2-d^2==0$

- Derivation: Algebraic expansion

- Basis: $\frac{\sqrt{a+b\,z}}{z\,\sqrt{c+d\,z}}==\frac{b\,c-a\,d}{c\,\sqrt{a+b\,z}\,\sqrt{c+d\,z}}+\frac{a\,\sqrt{c+d\,z}}{c\,z\,\sqrt{a+b\,z}}$

- Rule: If $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,c^2-d^2==0$, then

$$\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\text{Sin}[e+f\,x]\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx\,\rightarrow\,\frac{b\,c-a\,d}{c}\int \frac{1}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx+\frac{a}{c}\int \frac{\sqrt{c+d\,\text{Sin}[e+f\,x]}}{\text{Sin}[e+f\,x]\,\sqrt{a+b\,\text{Sin}[e+f\,x]}}\,dx$$

- Program code:

```
Int[Sqrt[a_+b_.*sin[e_.+f_.*x_]]/(sin[e_.+f_.*x_]*Sqrt[c_+d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  (b*c-a*d)/c*Int[1/(Sqrt[a+b*Sin[e+f*x]]*Sqrt[c+d*Sin[e+f*x]]),x] +
  a/c*Int[Sqrt[c+d*Sin[e+f*x]]/(Sin[e+f*x]*Sqrt[a+b*Sin[e+f*x]]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && EqQ[c^2-d^2,0]
```

2: $\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\text{Sin}[e+f\,x]\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,c^2-d^2\neq 0$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,c^2-d^2\neq 0$, then

$$\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\text{Sin}[e+f\,x]\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx\,\rightarrow$$

$$-\frac{2\,(a+b\,\mathrm{Sin}[e+f\,x])}{c\,f\,\sqrt{\frac{a+b}{c+d}}\,\mathrm{Cos}[e+f\,x]}\sqrt{-\frac{(b\,c-a\,d)\,(1-\mathrm{Sin}[e+f\,x])}{(c+d)\,(a+b\,\mathrm{Sin}[e+f\,x])}}$$

$$\sqrt{\frac{(b\,c-a\,d)\,(1+\mathrm{Sin}[e+f\,x])}{(c-d)\,(a+b\,\mathrm{Sin}[e+f\,x])}}\,\mathrm{EllipticPi}\left[\frac{a\,(c+d)}{c\,(a+b)},\,\mathrm{ArcSin}\left[\sqrt{\frac{a+b}{c+d}}\,\frac{\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}\right],\,\frac{(a-b)\,(c+d)}{(a+b)\,(c-d)}\right]$$

–

Program code:

```
Int[Sqrt[a_+b_.*sin[e_.+f_.*x_]]/(sin[e_.+f_.*x_]*Sqrt[c_+d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  -2*(a+b*Sin[e+f*x])/(c*f*Rt[(a+b)/(c+d),2]*Cos[e+f*x])*
    Sqrt[-(b*c-a*d)*(1-Sin[e+f*x])/((c+d)*(a+b*Sin[e+f*x]))]*Sqrt[(b*c-a*d)*(1+Sin[e+f*x])/((c-d)*(a+b*Sin[e+f*x]))]*
    EllipticPi[a*(c+d)/(c*(a+b)),ArcSin[Rt[(a+b)/(c+d),2]*Sqrt[c+d*Sin[e+f*x]]/Sqrt[a+b*Sin[e+f*x]]],(a-b)*(c+d)/((a+b)*(c-d))] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

2. $\int \frac{1}{\sin[e+f x] \sqrt{a+b \sin[e+f x]} \sqrt{c+d \sin[e+f x]}} \, dx$ when $b c-a d \neq 0$

1: $\int \frac{1}{\sin[e+f x] \sqrt{a+b \sin[e+f x]} \sqrt{c+d \sin[e+f x]}} \, dx$ when $b c-a d \neq 0 \wedge a^2-b^2=0 \wedge c^2-d^2=0$

Derivation: Piecewise constant extraction

Basis: If $a^2-b^2=0 \wedge c^2-d^2=0$, then $\partial_x \frac{\cos[e+f x]}{\sqrt{a+b \sin[e+f x]} \sqrt{c+d \sin[e+f x]}}=0$

Rule: If $b c-a d \neq 0 \wedge a^2-b^2=0 \wedge c^2-d^2=0$, then

$$\int \frac{1}{\sin[e+f x] \sqrt{a+b \sin[e+f x]} \sqrt{c+d \sin[e+f x]}} \, dx \rightarrow \frac{\cos[e+f x]}{\sqrt{a+b \sin[e+f x]} \sqrt{c+d \sin[e+f x]}} \int \frac{1}{\cos[e+f x] \sin[e+f x]} \, dx$$

Program code:

```
Int[1/(sin[e_.+f_.*x_]*Sqrt[a_+b_.*sin[e_.+f_.*x_]]*Sqrt[c_+d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  Cos[e+f*x]/(Sqrt[a+b*Sin[e+f*x]]*Sqrt[c+d*Sin[e+f*x]])*Int[1/(Cos[e+f*x]*Sin[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && EqQ[c^2-d^2,0]
```

2: $\int \frac{1}{\sin[e+f x] \sqrt{a+b \sin[e+f x]} \sqrt{c+d \sin[e+f x]}} \, dx$ when $b c-a d \neq 0 \wedge \left(a^2-b^2 \neq 0 \vee c^2-d^2 \neq 0\right)$

Derivation: Algebraic expansion

Basis: $\frac{1}{z \sqrt{a+b z}}=-\frac{b}{a \sqrt{a+b z}}+\frac{\sqrt{a+b z}}{a z}$

Rule: If $b c-a d \neq 0 \wedge \left(a^2-b^2 \neq 0 \vee c^2-d^2 \neq 0\right)$, then

$$\int \frac{1}{\mathrm{Sin}[e+f\,x]\,\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx \to -\frac{b}{a}\int \frac{1}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx + \frac{1}{a}\int \frac{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}{\mathrm{Sin}[e+f\,x]\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx$$

Program code:

```
Int[1/(sin[e_.+f_.*x_]*Sqrt[a_+b_.*sin[e_.+f_.*x_]]*Sqrt[c_+d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  -b/a*Int[1/(Sqrt[a+b*Sin[e+f*x]]*Sqrt[c+d*Sin[e+f*x]]),x] +
  1/a*Int[Sqrt[a+b*Sin[e+f*x]]/(Sin[e+f*x]*Sqrt[c+d*Sin[e+f*x]]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && (NeQ[a^2-b^2,0] || NeQ[c^2-d^2,0])
```

3. $\int \frac{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}{\mathrm{Sin}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0$

1: $\int \frac{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}{\mathrm{Sin}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0\wedge a^2-b^2==0\wedge c^2-d^2==0$

Derivation: Piecewise constant extraction

Basis: If $a^2-b^2==0\wedge c^2-d^2==0$, then $\partial_x \frac{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}{\mathrm{Cos}[e+f\,x]}==0$

Rule: If $b\,c-a\,d\neq 0\wedge a^2-b^2==0\wedge c^2-d^2==0$, then

$$\int \frac{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}{\mathrm{Sin}[e+f\,x]}\,dx \to \frac{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}{\mathrm{Cos}[e+f\,x]}\int \mathrm{Cot}[e+f\,x]\,dx$$

Program code:

```
Int[Sqrt[a_+b_.*sin[e_.+f_.*x_]]*Sqrt[c_+d_.*sin[e_.+f_.*x_]]/sin[e_.+f_.*x_],x_Symbol] :=
  Sqrt[a+b*Sin[e+f*x]]*Sqrt[c+d*Sin[e+f*x]]/Cos[e+f*x]*Int[Cot[e+f*x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && EqQ[c^2-d^2,0]
```

2: $\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}{\text{Sin}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0\,\land\,\left(a^2-b^2\neq 0\,\lor\,c^2-d^2\neq 0\right)$

Derivation: Algebraic expansion

Basis: $\frac{\sqrt{c+d\,z}}{z} == \frac{d}{\sqrt{c+d\,z}}+\frac{c}{z\,\sqrt{c+d\,z}}$

Rule: If $b\,c-a\,d\neq 0\,\land\,\left(a^2-b^2\neq 0\,\lor\,c^2-d^2\neq 0\right)$, then

$$\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}{\text{Sin}[e+f\,x]}\,dx \longrightarrow d\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx + c\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\text{Sin}[e+f\,x]\,\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx$$

Program code:

```
Int[Sqrt[a_+b_.*sin[e_.+f_.*x_]]*Sqrt[c_+d_.*sin[e_.+f_.*x_]]/sin[e_.+f_.*x_],x_Symbol] :=
  d*Int[Sqrt[a+b*Sin[e+f*x]]/Sqrt[c+d*Sin[e+f*x]],x] +
  c*Int[Sqrt[a+b*Sin[e+f*x]]/(Sin[e+f*x]*Sqrt[c+d*Sin[e+f*x]]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && (NeQ[a^2-b^2,0] || NeQ[c^2-d^2,0])
```

3: $\int \text{Sin}[e+f x]^p (a+b\,\text{Sin}[e+f x])^m (c+d\,\text{Sin}[e+f x])^n \,dx$ when $b c+a d==0 \wedge a^2-b^2==0 \wedge p+2 n==0 \wedge n\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $b c+a d==0 \wedge a^2-b^2==0 \wedge p+2 n==0 \wedge n\in\mathbb{Z}$, then
$\text{Sin}[e+f x]^p (c+d\,\text{Sin}[e+f x])^n == a^n c^n \text{Tan}[e+f x]^p (a+b\,\text{Sin}[e+f x])^{-n}$

Rule: If $b c+a d==0 \wedge a^2-b^2==0 \wedge p+2 n==0 \wedge n\in\mathbb{Z}$, then

$$\int \text{Sin}[e+f x]^p (a+b\,\text{Sin}[e+f x])^m (c+d\,\text{Sin}[e+f x])^n \,dx \to a^n c^n \int \text{Tan}[e+f x]^p (a+b\,\text{Sin}[e+f x])^{m-n} \,dx$$

Program code:

```
Int[sin[e_.+f_.*x_]^p_*(a_+b_.*sin[e_.+f_.*x_])^m_.*(c_+d_.*sin[e_.+f_.*x_])^n_.,x_Symbol] :=
  a^n*c^n*Int[Tan[e+f*x]^p*(a+b*Sin[e+f*x])^(m-n),x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && EqQ[p+2*n,0] && IntegerQ[n]
```

4: $\int (g\,\text{Sin}[e+f x])^p (a+b\,\text{Sin}[e+f x])^m (c+d\,\text{Sin}[e+f x])^n \,dx$ when $b c-a d\neq 0 \wedge a^2-b^2==0 \wedge c^2-d^2\neq 0 \wedge m-\frac{1}{2}\in\mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

Basis: If $a^2-b^2==0$, then $\partial_x \frac{\sqrt{a-b\,\text{Sin}[e+f x]}\,\sqrt{a+b\,\text{Sin}[e+f x]}}{\text{Cos}[e+f x]}==0$

Basis: $\text{Cos}[e+f x]==\frac{1}{f}\partial_x \text{Sin}[e+f x]$

Rule: If $b c-a d\neq 0 \wedge a^2-b^2==0 \wedge c^2-d^2\neq 0 \wedge m-\frac{1}{2}\in\mathbb{Z}$, then

$$\int (g\,\text{Sin}[e+f x])^p (a+b\,\text{Sin}[e+f x])^m (c+d\,\text{Sin}[e+f x])^n \,dx \to$$

$$\frac{\sqrt{a-b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}{\mathrm{Cos}[e+f\,x]}\int\frac{\mathrm{Cos}[e+f\,x]\,(g\,\mathrm{Sin}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^{m-\frac{1}{2}}\,(c+d\,\mathrm{Sin}[e+f\,x])^n}{\sqrt{a-b\,\mathrm{Sin}[e+f\,x]}}\,dx\,\longrightarrow$$

$$\frac{\sqrt{a-b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}{f\,\mathrm{Cos}[e+f\,x]}\,\mathrm{Subst}\left[\int\frac{(g\,x)^p\,(a+b\,x)^{m-\frac{1}{2}}\,(c+d\,x)^n}{\sqrt{a-b\,x}}\,dx,\,x,\,\mathrm{Sin}[e+f\,x]\right]$$

Program code:

```
Int[(g_.*sin[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  Sqrt[a-b*Sin[e+f*x]]*Sqrt[a+b*Sin[e+f*x]]/(f*Cos[e+f*x])*
    Subst[Int[(g*x)^p*(a+b*x)^(m-1/2)*(c+d*x)^n/Sqrt[a-b*x],x],x,Sin[e+f*x]] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && IntegerQ[m-1/2]
```

5: $\int(g\,\mathrm{Sin}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,((m\mid n)\in\mathbb{Z}\,\vee\,(m\mid p)\in\mathbb{Z}\,\vee\,(n\mid p)\in\mathbb{Z})$

- Derivation: Algebraic expansion

- Note: If p equal 1 or 2, better to use rules for integrands of the form $(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,(A+B\,\mathrm{Sin}[e+f\,x])$ or $(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,(A+B\,\mathrm{Sin}[e+f\,x]+C\,\mathrm{Sin}[e+f\,x]^2)$ respectively.

- Rule: If $b\,c-a\,d\neq 0\,\wedge\,((m\mid n)\in\mathbb{Z}\,\vee\,(m\mid p)\in\mathbb{Z}\,\vee\,(n\mid p)\in\mathbb{Z})$, then

$$\int(g\,\mathrm{Sin}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx\,\longrightarrow$$

$$\int\mathrm{ExpandTrig}\left[(g\,\mathrm{Sin}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n,\,x\right]dx$$

Program code:

```
Int[(g_.*sin[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  Int[ExpandTrig[(g*sin[e+f*x])^p*(a+b*sin[e+f*x])^m*(c+d*sin[e+f*x])^n,x],x] /;
FreeQ[{a,b,c,d,e,f,g,n,p},x] && NeQ[b*c-a*d,0] && (IntegersQ[m,n] || IntegersQ[m,p] || IntegersQ[n,p]) && NeQ[p,2]
```

X: $\int (g \operatorname{Sin}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Sin}[e+f x])^n \, dx$

Rule:

$$\int (g \operatorname{Sin}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Sin}[e+f x])^n \, dx \to \int (g \operatorname{Sin}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Sin}[e+f x])^n \, dx$$

Program code:

```
Int[(g_.*sin[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  Unintegrable[(g*Sin[e+f*x])^p*(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x] && NeQ[p,2]
```

Rules for integrands of the form $(g\,\text{Sin}[e+f\,x])^p\,(a+b\,\text{Csc}[e+f\,x])^m\,(c+d\,\text{Csc}[e+f\,x])^n$

1. $\int (g\,\text{Sin}[e+f\,x])^p\,(a+b\,\text{Csc}[e+f\,x])^m\,(c+d\,\text{Csc}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\land\,p\notin\mathbb{Z}$

   1: $\int (g\,\text{Sin}[e+f\,x])^p\,(a+b\,\text{Csc}[e+f\,x])^m\,(c+d\,\text{Csc}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\land\,p\notin\mathbb{Z}\,\land\,m\in\mathbb{Z}\,\land\,n\in\mathbb{Z}$

Derivation: Algebraic normalization

Basis: $a+b\,\text{Csc}[z]==\frac{b+a\,\text{Sin}[z]}{\text{Sin}[z]}$

Rule: If $b\,c-a\,d\neq 0\,\land\,p\notin\mathbb{Z}\,\land\,m\in\mathbb{Z}\,\land\,n\in\mathbb{Z}$, then

$$\int (g\,\text{Sin}[e+f\,x])^p\,(a+b\,\text{Csc}[e+f\,x])^m\,(c+d\,\text{Csc}[e+f\,x])^n\,dx\,\rightarrow\,g^{m+n}\int (g\,\text{Sin}[e+f\,x])^{p-m-n}\,(b+a\,\text{Sin}[e+f\,x])^m\,(d+c\,\text{Sin}[e+f\,x])^n\,dx$$

Program code:

```
Int[(g_.*sin[e_.+f_.*x_])^p_.*(a_.+b_.*csc[e_.+f_.*x_])^m_.*(c_+d_.*csc[e_.+f_.*x_])^n_.,x_Symbol] :=
  g^(m+n)*Int[(g*Sin[e+f*x])^(p-m-n)*(b+a*Sin[e+f*x])^m*(d+c*Sin[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,p},x] && NeQ[b*c-a*d,0] && Not[IntegerQ[p]] && IntegerQ[m] && IntegerQ[n]
```

2: $\int (g \sin[e+f x])^p (a+b \csc[e+f x])^m (c+d \csc[e+f x])^n \, dx$ when $b c-a d \neq 0 \land p \notin \mathbb{Z} \land \neg (m \in \mathbb{Z} \land n \in \mathbb{Z})$

Derivation: Piecewise constant extraction

Basis: $\partial_x \left((g \cos[e+f x])^p (g \sec[e+f x])^p\right) == 0$

Rule: If $b c-a d \neq 0 \land p \notin \mathbb{Z} \land \neg (m \in \mathbb{Z} \land n \in \mathbb{Z})$, then

$$\int (g \sin[e+f x])^p (a+b \csc[e+f x])^m (c+d \csc[e+f x])^n \, dx \longrightarrow (g \csc[e+f x])^p (g \sin[e+f x])^p \int \frac{(a+b \csc[e+f x])^m (c+d \csc[e+f x])^n}{(g \csc[e+f x])^p} \, dx$$

–

Program code:

```
Int[(g_.*sin[e_.+f_.*x_])^p_.*(a_.+b_.*csc[e_.+f_.*x_])^m_.*(c_+d_.*csc[e_.+f_.*x_])^n_.,x_Symbol] :=
  (g*Csc[e+f*x])^p*(g*Sin[e+f*x])^p*Int[(a+b*Csc[e+f*x])^m*(c+d*Csc[e+f*x])^n/(g*Csc[e+f*x])^p,x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x] && NeQ[b*c-a*d,0] && Not[IntegerQ[p]] && Not[IntegerQ[m] && IntegerQ[n]]
```

Rules for integrands of the form $(g\,\text{Sin}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Csc}[e+f\,x])^n$

1: $\int (g\,\text{Sin}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Csc}[e+f\,x])^n\,dx$ when $n\in\mathbb{Z}$

Derivation: Algebraic normalization

Basis: $c+d\,\text{Csc}[z]==\frac{d+c\,\text{Sin}[z]}{\text{Sin}[z]}$

Rule: If $n\in\mathbb{Z}$, then

$$\int (g\,\text{Sin}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Csc}[e+f\,x])^n\,dx\;\longrightarrow\;g^n\int (g\,\text{Sin}[e+f\,x])^{p-n}\,(a+b\,\text{Sin}[e+f\,x])^m\,(d+c\,\text{Sin}[e+f\,x])^n\,dx$$

Program code:

```
Int[(g_.*sin[e_.+f_.*x_])^p_.*(a_+b_.*sin[e_.+f_.*x_])^m_.*(c_+d_.*csc[e_.+f_.*x_])^n_.,x_Symbol] :=
  g^n*Int[(g*Sin[e+f*x])^(p-n)*(a+b*Sin[e+f*x])^m*(d+c*Sin[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,m,p},x] && IntegerQ[n]
```

2. $\int (g \operatorname{Sin}[e+fx])^p (a+b \operatorname{Sin}[e+fx])^m (c+d \operatorname{Csc}[e+fx])^n \, dx$ when $n \notin \mathbb{Z}$

1. $\int (g \operatorname{Sin}[e+fx])^p (a+b \operatorname{Sin}[e+fx])^m (c+d \operatorname{Csc}[e+fx])^n \, dx$ when $n \notin \mathbb{Z} \land m \in \mathbb{Z}$

1: $\int \operatorname{Sin}[e+fx]^p (a+b \operatorname{Sin}[e+fx])^m (c+d \operatorname{Csc}[e+fx])^n \, dx$ when $n \notin \mathbb{Z} \land m \in \mathbb{Z} \land p \in \mathbb{Z}$

Derivation: Algebraic normalization

Basis: $a+b \operatorname{Sin}[z] == \frac{b+a \operatorname{Csc}[z]}{\operatorname{Csc}[z]}$

Rule: If $n \notin \mathbb{Z} \land m \in \mathbb{Z} \land p \in \mathbb{Z}$, then

$$\int \operatorname{Sin}[e+fx]^p (a+b \operatorname{Sin}[e+fx])^m (c+d \operatorname{Csc}[e+fx])^n \, dx \longrightarrow \int \frac{(b+a \operatorname{Csc}[e+fx])^m (c+d \operatorname{Csc}[e+fx])^n}{\operatorname{Csc}[e+fx]^{m+p}} \, dx$$

Program code:

```
Int[sin[e_.+f_.*x_]^p_.*(a_+b_.*sin[e_.+f_.*x_])^m_.*(c_+d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  Int[(b+a*Csc[e+f*x])^m*(c+d*Csc[e+f*x])^n/Csc[e+f*x]^(m+p),x] /;
FreeQ[{a,b,c,d,e,f,n},x] && Not[IntegerQ[n]] && IntegerQ[m] && IntegerQ[p]
```

2: $\int (g \operatorname{Sin}[e+fx])^p (a+b \operatorname{Sin}[e+fx])^m (c+d \operatorname{Csc}[e+fx])^n \, dx$ when $n \notin \mathbb{Z} \land m \in \mathbb{Z} \land p \notin \mathbb{Z}$

Derivation: Algebraic normalization and piecewise constant extraction

Basis: $a+b \operatorname{Sin}[z] == \frac{b+a \operatorname{Csc}[z]}{\operatorname{Csc}[z]}$

Basis: $\partial_x \left(\operatorname{Csc}[e+fx]^p (g \operatorname{Sin}[e+fx])^p\right) == 0$

Rule: If $n \notin \mathbb{Z} \land m \in \mathbb{Z} \land p \notin \mathbb{Z}$, then

$$\int (g \operatorname{Sin}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Csc}[e+f x])^n \, dx \to \operatorname{Csc}[e+f x]^p (g \operatorname{Sin}[e+f x])^p \int \frac{(b+a \operatorname{Csc}[e+f x])^m (c+d \operatorname{Csc}[e+f x])^n}{\operatorname{Csc}[e+f x]^{m+p}} \, dx$$

Program code:

```
Int[(g_.*sin[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_.*(c_+d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  Csc[e+f*x]^p*(g*Sin[e+f*x])^p*Int[(b+a*Csc[e+f*x])^m*(c+d*Csc[e+f*x])^n/Csc[e+f*x]^(m+p),x] /;
FreeQ[{a,b,c,d,e,f,g,n,p},x] && Not[IntegerQ[n]] && IntegerQ[m] && Not[IntegerQ[p]]
```

2: $\int (g \operatorname{Sin}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Csc}[e+f x])^n \, dx$ when $n \notin \mathbb{Z} \land m \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(g \operatorname{Sin}[e+f x])^n (c+d \operatorname{Csc}[e+f x])^n}{(d+c \operatorname{Sin}[e+f x])^n} == 0$

Rule: If $n \notin \mathbb{Z} \land m \notin \mathbb{Z}$, then

$$\int (g \operatorname{Sin}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Csc}[e+f x])^n \, dx \to$$
$$\frac{(g \operatorname{Sin}[e+f x])^n (c+d \operatorname{Csc}[e+f x])^n}{(d+c \operatorname{Sin}[e+f x])^n} \int (g \operatorname{Sin}[e+f x])^{p-n} (a+b \operatorname{Sin}[e+f x])^m (d+c \operatorname{Sin}[e+f x])^n \, dx$$

Program code:

```
Int[(g_.*sin[e_.+f_.*x_])^p_.*(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  (g*Sin[e+f*x])^n*(c+d*Csc[e+f*x])^n/(d+c*Sin[e+f*x])^n*Int[(g*Sin[e+f*x])^(p-n)*(a+b*Sin[e+f*x])^m*(d+c*Sin[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x] && Not[IntegerQ[n]] && Not[IntegerQ[m]]
```

Rules for integrands of the form $(g\,\text{Csc}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n$

1. $\int (g\,\text{Csc}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,p\notin\mathbb{Z}$

1: $\int (g\,\text{Csc}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,m\in\mathbb{Z}\,\wedge\,n\in\mathbb{Z}$

Derivation: Algebraic normalization

Basis: $a+b\,\text{Sin}[z]==\frac{b+a\,\text{Csc}[z]}{\text{Csc}[z]}$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,m\in\mathbb{Z}\,\wedge\,n\in\mathbb{Z}$, then

$$\int (g\,\text{Csc}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx\,\longrightarrow\,g^{m+n}\int (g\,\text{Csc}[e+f\,x])^{p-m-n}\,(b+a\,\text{Csc}[e+f\,x])^m\,(d+c\,\text{Csc}[e+f\,x])^n\,dx$$

Program code:

```
Int[(g_.*csc[e_.+f_.*x_])^p_.*(a_.+b_.*sin[e_.+f_.*x_])^m_.*(c_+d_.*sin[e_.+f_.*x_])^n_.,x_Symbol] :=
  g^(m+n)*Int[(g*Csc[e+f*x])^(p-m-n)*(b+a*Csc[e+f*x])^m*(d+c*Csc[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,p},x] && NeQ[b*c-a*d,0] && Not[IntegerQ[p]] && IntegerQ[m] && IntegerQ[n]
```

2: $\int (g \operatorname{Csc}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Sin}[e+f x])^n \, dx$ when $b c-a d \neq 0 \land p \notin \mathbb{Z} \land \neg (m \in \mathbb{Z} \land n \in \mathbb{Z})$

Derivation: Piecewise constant extraction

Basis: $\partial_x \left((g \operatorname{Csc}[e+f x])^p (g \operatorname{Sin}[e+f x])^p \right) = 0$

Rule: If $b c-a d \neq 0 \land p \notin \mathbb{Z} \land \neg (m \in \mathbb{Z} \land n \in \mathbb{Z})$, then

$$\int (g \operatorname{Csc}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Sin}[e+f x])^n \, dx \to (g \operatorname{Csc}[e+f x])^p (g \operatorname{Sin}[e+f x])^p \int \frac{(a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Sin}[e+f x])^n}{(g \operatorname{Sin}[e+f x])^p} \, dx$$

–

Program code:

```
Int[(g_.*csc[e_.+f_.*x_])^p_.*(a_.+b_.*sin[e_.+f_.*x_])^m_.*(c_+d_.*sin[e_.+f_.*x_])^n_.,x_Symbol] :=
  (g*Csc[e+f*x])^p*(g*Sin[e+f*x])^p*Int[(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^n/(g*Sin[e+f*x])^p,x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x] && NeQ[b*c-a*d,0] && Not[IntegerQ[p]] && Not[IntegerQ[m] && IntegerQ[n]]
```

Rules for integrands of the form $(g\,\text{Csc}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Csc}[e+f\,x])^n$

1: $\int (g\,\text{Csc}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Csc}[e+f\,x])^n\,dx$ when $m\in\mathbb{Z}$

Derivation: Algebraic normalization

Basis: $a+b\,\text{Sin}[z]==\frac{b+a\,\text{Csc}[z]}{\text{Csc}[z]}$

Rule: If $m\in\mathbb{Z}$, then

$$\int (g\,\text{Csc}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Csc}[e+f\,x])^n\,dx\;\longrightarrow\;g^m\int (g\,\text{Csc}[e+f\,x])^{p-m}\,(b+a\,\text{Csc}[e+f\,x])^m\,(c+d\,\text{Csc}[e+f\,x])^n\,dx$$

Program code:

```
Int[(g_.*csc[e_.+f_.*x_])^p_.*(a_+b_.*sin[e_.+f_.*x_])^m_.*(c_+d_.*csc[e_.+f_.*x_])^n_.,x_Symbol] :=
  g^m*Int[(g*Csc[e+f*x])^(p-m)*(b+a*Csc[e+f*x])^m*(c+d*Csc[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,n,p},x] && IntegerQ[m]
```

2. $\int (g \operatorname{Csc}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Csc}[e+f x])^n \, dx$ when $m \notin \mathbb{Z}$

   1. $\int (g \operatorname{Csc}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Csc}[e+f x])^n \, dx$ when $m \notin \mathbb{Z} \land n \in \mathbb{Z}$

      1: $\int \operatorname{Csc}[e+f x]^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Csc}[e+f x])^n \, dx$ when $m \notin \mathbb{Z} \land n \in \mathbb{Z} \land p \in \mathbb{Z}$

Derivation: Algebraic normalization

Basis: $c + d \operatorname{Csc}[z] == \frac{d + c \operatorname{Sin}[z]}{\operatorname{Sin}[z]}$

Rule: If $m \notin \mathbb{Z} \land n \in \mathbb{Z} \land p \in \mathbb{Z}$, then

$$\int \operatorname{Csc}[e+f x]^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Csc}[e+f x])^n \, dx \longrightarrow \int \frac{(a+b \operatorname{Sin}[e+f x])^m (d+c \operatorname{Sin}[e+f x])^n}{\operatorname{Sin}[e+f x]^{n+p}} \, dx$$

Program code:

```
Int[csc[e_.+f_.*x_]^p_.*(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*csc[e_.+f_.*x_])^n_.,x_Symbol] :=
  Int[(a+b*Sin[e+f*x])^m*(d+c*Sin[e+f*x])^n/Sin[e+f*x]^(n+p),x] /;
FreeQ[{a,b,c,d,e,f,m},x] && Not[IntegerQ[m]] && IntegerQ[n] && IntegerQ[p]
```

      2: $\int (g \operatorname{Csc}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Csc}[e+f x])^n \, dx$ when $m \notin \mathbb{Z} \land n \in \mathbb{Z} \land p \notin \mathbb{Z}$

Derivation: Algebraic normalization and piecewise constant extraction

Basis: $c + d \operatorname{Csc}[z] == \frac{d + c \operatorname{Sin}[z]}{\operatorname{Sin}[z]}$

Basis: $\partial_x \left(\operatorname{Sin}[e+f x]^p (g \operatorname{Csc}[e+f x])^p\right) == 0$

Rule: If $m \notin \mathbb{Z} \land n \in \mathbb{Z} \land p \notin \mathbb{Z}$, then

$$\int (g\,\mathrm{Csc}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Csc}[e+f\,x])^n\,dx \longrightarrow \mathrm{Sin}[e+f\,x]^p\,(g\,\mathrm{Csc}[e+f\,x])^p\int\frac{(a+b\,\mathrm{Sin}[e+f\,x])^m\,(d+c\,\mathrm{Sin}[e+f\,x])^n}{\mathrm{Sin}[e+f\,x]^{n+p}}\,dx$$

Program code:

```
Int[(g_.*csc[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*csc[e_.+f_.*x_])^n_.,x_Symbol] :=
  Sin[e+f*x]^p*(g*Csc[e+f*x])^p*Int[(a+b*Sin[e+f*x])^m*(d+c*Sin[e+f*x])^n/Sin[e+f*x]^(n+p),x] /;
FreeQ[{a,b,c,d,e,f,g,m,p},x] && Not[IntegerQ[m]] && IntegerQ[n] && Not[IntegerQ[p]]
```

2: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Csc}[e+f\,x])^n\,(g\,\mathrm{Csc}[e+f\,x])^p\,dx$ when $m\notin\mathbb{Z}\land n\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x\frac{(g\,\mathrm{Csc}[e+f\,x])^m\,(a+b\,\mathrm{Sin}[e+f\,x])^m}{(b+a\,\mathrm{Csc}[e+f\,x])^m}==0$

Rule: If $m\notin\mathbb{Z}\land n\notin\mathbb{Z}$, then

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Csc}[e+f\,x])^n\,(g\,\mathrm{Csc}[e+f\,x])^p\,dx \longrightarrow$$
$$\frac{(a+b\,\mathrm{Sin}[e+f\,x])^m\,(g\,\mathrm{Csc}[e+f\,x])^m}{(b+a\,\mathrm{Csc}[e+f\,x])^m}\int (g\,\mathrm{Csc}[e+f\,x])^{p-m}\,(b+a\,\mathrm{Csc}[e+f\,x])^m\,(c+d\,\mathrm{Csc}[e+f\,x])^n\,dx$$

Program code:

```
Int[(g_.*csc[e_.+f_.*x_])^p_.*(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  (a+b*Sin[e+f*x])^m*(g*Csc[e+f*x])^m/(b+a*Csc[e+f*x])^m*
    Int[(g*Csc[e+f*x])^(p-m)*(b+a*Csc[e+f*x])^m*(c+d*Csc[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x] && Not[IntegerQ[m]] && Not[IntegerQ[n]]
```